Пример программных модулей (на Delphi) для Типового расчета
для решения задачи со стеком (LIFO), смоделированным на основе 
а) односвязного линейного динамического списка;
б) одномерного динамического массива.
Стек – последовательность однородных элементов с порядком обработки LIFO (Last In First Out – последний пришедший удаляется первым).
Часть 1.
Для работы со стеком надо определить тип TStack (на основе двух разных базовых типов) и реализовать несколько элементарных операций:
- проверка стека на пустоту (IsEmpty);
- установка начальных значений (InitStack);
- добавление элемента (AddTop) в вершину стека;
- удаление элемента (DelTop) из вершины стека;
- запрос информационной части элемента из вершины стека (GetTop);
и несколько вспомогательных, но часто встречающихся операций:
- очистка стека (FreeStack) с освобождением занимаемой памяти;
- перекладывание элемента из одного стека в другой (TopToTop).
а) На основе линейного односвязного СПИСКА:
 на основе ссылочного типа «Линейный односвязный список»;

unit UnStList;
interface
Const
  MaxLength = 10;
Type
  TInfo=String[MaxLength];
  PElem=^TElem;
  TElem=record
          Info: TInfo;
          Next: PElem;
  End;
  TStack = PElem; // моделирование стека на основе односвязного списка
// пуст ли стек?
function IsEmpty(aStack: TStack): boolean;
// начальное значение (после описания переменной)
procedure InitStack(var aStack: TStack);
// добавление элемента в вершину стека
procedure AddTop(var aStack: TStack; aData: TInfo);
// узнать информационную часть вершины НЕпустого стека
function GetTop(aStack: TStack): TInfo;
// удаление элемента из вершины НЕпустого стека
procedure DelTop(var aStack: TStack);
// удаление всех элементов стека
procedure FreeStack(var aStack: TStack);
// процедура перекладывания элемента из одного стека в другой
procedure TopToTop(var aStack1, aStack2: TStack);
//---------------------------------------------------------
implementation
function IsEmpty(aStack: TStack): boolean;
// пуст ли стек?
begin
  // стек пуст, если адрес вершины стека = Nil
  IsEmpty := aStack = Nil
end;
procedure InitStack(var aStack: TStack);
// начальное значение
begin
  aStack := Nil
end;
procedure AddTop(var aStack: TStack; aData: TInfo);
// добавление элемента в вершину стека
var NewElem: TStack;
begin
  new(NewElem);  // выделяем память и заполняем поля нового элемента
  NewElem^.Info := aData;
  NewElem^.Next := aStack;
  aStack := NewElem; //новый элемент – новая вершина
end;
function GetTop(aStack: TStack): TInfo;
// информационная часть вершины НЕпустого стека (проверять пустоту заранее)
begin
  GetTop:=aStack^.Info
end;
procedure DelTop(var aStack: TStack);
// удаление элемента из вершины НЕпустого стека (проверять пустоту заранее)
var DelElem: TStack;
begin
  DelElem := aStack; // запоминаем адрес удаляемого элемента
  aStack := aStack^.Next; // новая вершина – следующий элемент
  Dispose(DelElem); // удаление (освобождение памяти)
end;
procedure FreeStack(var aStack: TStack);
// удаление всех элементов стека
begin
  while not IsEmpty(aStack) do // пока стек не пуст, удаляем вершину
    DelTop(aStack);
end;
procedure TopToTop(var aStack1, aStack2: TStack);
// процедура перекладывания элемента из одного стека в другой
// без освобождения памяти (только изменением связей)
var anElem: TStack;
begin
   anElem := aStack1;
   aStack1 := aStack1^.Next;
   anElem^.Next:= aStack2;
   aStack2:= anElem;
end;
end.
б) На основе одномерного динамического МАССИВА:
 (вершина стека в элементе с индексом Count-1, если стек пуст, то Count=0)

unit UnStMas;
interface
Const
  MaxLength = 10;
Type
  TInfo=String[MaxLength];
  TStack = record // моделирование стека на основе динамического массива
     Mas: array of TInfo;   // см. Рисунок выше
     Count, MaxCount : integer;
  end;
// пуст ли стек?
function IsEmpty(aStack: TStack): boolean;
// начальное значение после описания переменной
procedure InitStack(var aStack: TStack);
// добавление элемента в вершину стека
procedure AddTop(var aStack: TStack; aData: TInfo);
// узнать информационную часть вершины НЕпустого стека
function GetTop(aStack: TStack): TInfo;
// удаление элемента из вершины НЕпустого стека
procedure DelTop(var aStack: TStack);
// удаление всех элементов стека
procedure FreeStack(var aStack: TStack);
// процедура перекладывания элемента из одного стека в другой
procedure TopToTop(var aStack1, aStack2: TStack);
//---------------------------------------------------------
implementation
function IsEmpty(aStack: TStack): boolean;
// пуст ли стек?
begin
  // стек пуст, если адрес вершины стека = Nil
  IsEmpty := aStack.Count = 0
end;
procedure InitStack(var aStack: TStack);
// начальное значение для всех полей (после описания переменной стека)
begin
  with aStack do
  begin
    Count := 0;
    MaxCount :=0;
    Mas := nil;
  end;
end;
procedure AddTop(var aStack: TStack; aData: TInfo);
// добавление элемента в вершину стека = конец массива
begin
  with aStack do
  begin
    if Count=MaxCount then // если запаса нет, создаем его
    begin
      Inc(MaxCount,10);
      SetLength(Mas, MaxCount);  // выделяем с запасом - на 9 больше
    end;

// заполняем элемент в вершине (по индексу Count)
    Mas[Count] := aData; 
    Inc(Count);
  end
end;
function GetTop(aStack: TStack): TInfo;
// информационная часть вершины НЕпустого стека (проверять пустоту заранее)
begin
  GetTop:=aStack.Mas[aStack.Count-1]
end;
procedure DelTop(var aStack: TStack);
// удаление элемента из вершины НЕпустого стека (проверять пустоту заранее)
begin
  with aStack do
  begin
    Dec(Count);
    if Count < MaxCount - 20 then
    begin // освободим лишнюю память, но не всю, запас оставляем
      Dec(MaxCount,10);
      SetLength(Mas, MaxCount);
    end
  end
end;
procedure FreeStack(var aStack: TStack);
// удаление всех элементов стека
begin
  InitStack(aStack);
// Динамический массив - тип с управляемым временем жизни
// - память будет освобождена при присвоении Nil массиву
end;
procedure TopToTop(var aStack1, aStack2: TStack);
// процедура перекладывания элемента из одного стека в другой
var Info: TInfo;
begin
   Info:=GetTop(aStack1); // Pop = GetTop + DelTop
   DelTop(aStack1);
   AddTop(aStack2, Info);  // Push
end;
end.
Часть 2.
На основе описанного типа TStack и базовых операций, решить следующие подзадачи:
- очистить стек (Free);
- добавить элемент (AddToTop);
- удалить элемент (DeleteTop);
- решить задачу (Дан стек из строк длиной не более 10 символов, вычислить максимальную длину строк и удалить все строки максимальной длины из стека) (Decide);
- просмотреть содержимое стека (вывод на экран) (ViewStack).
Интерфейс
Для интерактивного взаимодействия с пользователем используется меню:
1 – Добавить
- добавить один элемент в стек
2 – Удалить 
- удалить один элемент из стека
3 – Решить 

- удалить строку(и) максимальной длины
4 – Просмотр 
- вывод информации из стека на экран
5 – Очистить 
- удалить все элементы из стека
6 - Выход
И написан программный код головного модуля:
Программный код – одинаковый для разных способов моделирования. Разница лишь в имени подключаемого модуля с одноименными типами и операциями со стеком:
Program TP;
{$Apptype CONSOLE}
uses UnStMas,  // либо uses UnStList; для способа №2

SysUtils, // там Trim – удаление пробелов в начале и конце строки

Windows; // для смены кодировки при консольном в/в
procedure AddToTop(var Stack: TStack);
// добавить в вершину
Var s: TInfo;
Begin
  Write('Введите:'); readln(s);
  s := Trim(s);
  if s<>'' then
  begin
    AddTop(Stack, s);
    writeln('Добавлена строка "', s,'"');
  end
  else
    writeln( 'Значение не введено, либо состоит только из пробелов'#13#10,
             'Не удалось добавить');
end;
procedure ViewStack(var Stack: TStack);
var DopStack: TStack;
begin
  InitStack(DopStack);
  Writeln('Стек:');
  if IsEmpty(Stack) then writeln('(пусто)');
  while not IsEmpty(Stack) do
  begin
     writeln(GetTop(Stack));
     TopToTop(Stack, DopStack);
  end;
  while not IsEmpty(DopStack) do
     TopToTop(DopStack, Stack);
end;
procedure DeleteTop(var Stack: TStack);
begin
  if not IsEmpty(Stack) then
  begin
    DelTop(Stack);
    ViewStack(Stack);
  end
  else
    writeln( 'Стек пуст - Не удалось удалить');
end;
procedure Free(var Stack: TStack);
begin
  FreeStack(Stack);
  ViewStack(Stack);
end;
procedure Decide(var Stack: TStack);
// удаление самого длинного слова, (всех длинных, если несколько одинаковой длины)
var
  MaxLen, aLen: byte;
  DopStack: TStack;
begin
  if IsEmpty(Stack) then
    writeln( 'Стек пуст - невозможно удалить самые длинные строки'#13#10,
             'Не удалось решить задачу')
  else
  begin
    InitStack(DopStack); // дополнительный стек, перекладывая элементы в него




// ищем максимум
    MaxLen := Length(GetTop(Stack));// нач. Значение максимума
    TopToTop(Stack, DopStack);
    while not IsEmpty(Stack) do
    begin
       aLen:=Length(GetTop(Stack));
       if aLen > MaxLen then MaxLen:= aLen; //если есть длиннее -> новый максимум
       TopToTop(Stack, DopStack);
    end;

// перекладывая элементы обратно, удаляем строки максимальной длины
    while not IsEmpty(DopStack) do
    begin
       if Length(GetTop(DopStack)) = MaxLen then
         DelTop(DopStack)
       else
         TopToTop(DopStack, Stack);
    end;
    writeln('Строки максимальной длины ', MaxLen, ' успешно удалены'#13#10,
            'Нажмите кнопку ПРОСМОТР, чтобы увидеть результат');
  end;
end;
//--------------------главная программа-------------------------
Var
  Stack: TStack;  // описание переменной
  ch: char;
begin
  SetConsoleCP(1251);      // из модуля Windows
  SetConsoleOutputCP(1251); // смена кодировки
  InitStack(Stack); // начальные значения
  repeat
      writeln('--------------------------------------');
      Writeln(' 1 – Добавить'#13#10, ' 2 - Удалить'#13#10, ' 3 - Решить'#13#10,


  ' 4 – Просмотр'#13#10, ' 5 - Очистить'#13#10, ' 6 - Выход');
      write('Ваш выбор?'); readln(ch);
      writeln('--------------------------------------');
      case ch of

  '1': AddToTop(Stack);

  '2': DeleteTop(Stack);

  '3': Decide(Stack);

  '4': ViewStack(Stack);

  '5': Free(Stack);

  '6': ;
        else
        begin
          writeln('Нет такой команды');
        end;
      end;
      if ch in ['1'..'4'] then begin write('Press ENTER'); readln; end;
  until ch='6';
end.
Заключение
Смоделирован новый тип «Стек» двумя способами: 1) на основе линейного односвязного списка и 2) на основе динамического одномерного массива.
1) В массив добавлять элементы (в вершину стека) несколько сложнее, чем в список, поскольку при этом приходится копировать все элементы массива в новую память большего размера. В приведенной реализации память перевыделяется не каждый раз при добавлении нового элемента, а создается запас из 9 элементов, что позволяет повысить эффективность не менее чем в 10 раз (в 10-20 раз: 20 – максимальный размер запаса, получающийся, например, после удаления 11 элементов при запасе 9). Остальные базовые операции примерно одинаковой сложности в обоих способах.
2) Количество памяти, выделяемое для хранения значений элементов стека, больше при использовании первого метода, поскольку хранятся, не только значения, но и ссылки между элементами, но при создании запаса элементов для облегчения процесса добавления элементов, этот запас при небольшом размере стека может сравниться с размером памяти для хранения ссылок:
Одна ссылка = 4 байта, один запасной элемент 11 байтов, то есть 
1 запасной элемент примерно равен 3 ссылкам. 
9 запасных элементов в массиве 9*11=99 примерно соответствует 25 ссылкам по 4 байта.
20 запасных элементов (максимальный размер запаса) 20*11=220 примерно соответствует 55 ссылкам в списке.
Таким образом, при размере массива более 55 элементов память для хранения ссылок в списке всегда превышает размер памяти «про запас» в массиве даже при максимальном запасе элементов в массиве, а при меньшей длине все зависит от текущего размера запаса.
Вывод: При использовании запаса памяти в массиве оба способа становятся почти равноценными (имеют и достоинства и недостатки) при небольшой длине (до 55 элементов) стека. Без использования запаса памяти в массиве – постоянное копирование всех элементов стека при каждом добавлении элемента в него нового элемента делает использование массива неэффективным по времени по сравнению со списком.
Примерные типы ОЧЕРЕДИ, СПИСКА, ДЕРЕВА бинарного, ДЕКа, МАТРИЦЫ:
Const
  MaxLength = 10;
Type
  TInfo=String[MaxLength]; // тип элемента зависит от задачи
  PElem=^TElem;
  TElem=record
          Info: TInfo;
          Next: PElem;  // количество связей может быть другим
  End;
// моделирование СТЕКА на основе односвязного списка
  TStack = PElem; 
// моделирование ОЧЕРЕДИ на основе односвязного списка
  TQueue = record 
      Head: PElem; 
      Tail: PElem; 
  End;
// моделирование ДЕКа на основе ДВУсвязного списка
  TDEQ = record 
      End1: PElem; 
      End2: PElem; 
      napravlenie: boolean; 
  End;
// моделирование МАТРИЦЫ на основе ДВУсвязного списка
  TMatrix = record 
      First: PElem; 
      Current: PElem; 
      n, m, i, j: byte;
  End;
// моделирование ДЕРЕВА бинарного на основе ТРЁХсвязного списка
  TTree = record 
      Root: PElem; 
      Current: PElem; 
  End;
 // моделирование СТЕКА на основе динамического МАССИВА
  TStack = record
     Mas: array of TInfo;   // см. Рисунок выше
     Count, MaxCount: integer;
  end;
 // моделирование СПИСКА на основе динамического МАССИВА
  TList = record
     Mas: array of TInfo;   
     First, Last, MaxCount, Cur : integer;
  end;
 // моделирование ДЕРЕВА бинарного на основе динамического МАССИВА
  TTree = record
     Mas: array of TInfo;   
     Current, Count, MaxCount : integer;
  end;
 // моделирование ДЕКа на основе динамического МАССИВА
  TDEQ = record
     Mas: array of TInfo;   // может быть два массива
     End1, End2, MaxCount : integer;
  end;
// моделирование ОЧЕРЕДИ на основе типизированного ФАЙЛА
  TQueue = record 
      FName: String; 
      F: File of TInfo;
      Head, Tail: integer; 
  End;
// моделирование СПИСКА на основе типизированного ФАЙЛА
  TLIst = record 
      FName: String; 
      F: File of TInfo;
      First, Last, Cur: integer; 
  End;
// моделирование ДЕРЕВА бинарного на основе типизированного ФАЙЛА
  TTree = record 
      FName: String; 
      F: File of TInfo;
      Current, Count: integer; 
  End;
// моделирование МАТРИЦЫ на основе типизированного ФАЙЛА
  TMatrix = record 
      FName: String; 
      F: File of TInfo;
      n, m, i, j: byte;
  End;

Запас – при добавлении не выгодно каждый раз выделять новую память и копировать в нее массив, эффективнее выделять память с запасом.

